

graph twoway rcap — Range plot with capped spikes[Description](#)
[Options](#)[Quick start](#)
[Remarks and examples](#)[Menu](#)
[Also see](#)[Syntax](#)

Description

A range plot has two y variables, such as high and low daily stock prices or upper and lower 95% confidence limits.

`twoway rcap` plots a range, using capped spikes (I-beams) to connect the high and low values.

Quick start

Range plot with capped spikes

```
twoway rcap y1 y2 x
```

Same as above, with horizontal spikes

```
twoway rcap y1 y2 x, horizontal
```

Specify small caps

```
twoway rcap y1 y2 x, msize(small)
```

Same as above, but specify orange spikes

```
twoway rcap y1 y2 x, msize(small) lcolor(orange)
```

Use approximately 5 tick marks at round values on the y axis and label them

```
twoway rcap y1 y2 x, ylabel(#5)
```

Menu

Graphics > Twoway graph (scatter, line, etc.)

Syntax

```
twoway rcap y1var y2var xvar [if] [in] [, options]
```

<i>options</i>	Description
<code>msize(<i>markersizestyle</i>)</code>	width of cap
<code>vertical</code>	vertical spikes; the default
<code>horizontal</code>	horizontal spikes
<code>line_options</code>	change look of spike and cap lines
<code>colorvar_options</code>	change color of spike and cap lines based on values of a variable
<code>axis_choice_options</code>	associate plot with alternative axis
<code>twoway_options</code>	titles, legends, axes, added lines and text, by, regions, name, aspect ratio, etc.

All explicit options are *rightmost*, except `vertical` and `horizontal`, which are *unique*; see [G-4] **Concept: repeated options**.

Options

`msize(markersizestyle)` specifies the width of the cap. Option `msize()` is in fact `twoway scatter`'s *marker_option* that sets the size of the marker symbol, but here `mymbol()` is borrowed to set the cap width. See [G-4] *markersizestyle* for a list of size choices.

`vertical` and `horizontal` specify whether the high and low *y* values are to be presented vertically (the default) or horizontally.

In the default `vertical` case, *y1var* and *y2var* record the minimum and maximum (or maximum and minimum) *y* values to be graphed against each *xvar* value.

If `horizontal` is specified, the values recorded in *y1var* and *y2var* are plotted in the *x* direction, and *xvar* is treated as the *y* value.

`line_options` specify the look of the lines used to draw the spikes and their caps, including pattern, width, and color; see [G-3] *line_options*.

`colorvar_options` specify that the color of the lines used to draw the spikes and their caps be determined by the levels of the numeric variable *colorvar*; see [G-3] *colorvar_options*.

`axis_choice_options` associate the plot with a particular *y* or *x* axis on the graph; see [G-3] *axis_choice_options*.

`twoway_options` are a set of common options supported by all `twoway` graphs. These options allow you to title graphs, name graphs, control axes and legends, add lines and text, set aspect ratios, create graphs over `by()` groups, and change some advanced settings. See [G-3] *twoway_options*.

Remarks and examples

Remarks are presented under the following headings:

Typical use
Advanced use
Advanced use 2

Typical use


We have daily data recording the values for the S&P 500 in 2001:

```
. use https://www.stata-press.com/data/r18/sp500
(S&P 500)
. list date high low close in 1/5
```

	date	high	low	close
1.	02jan2001	1320.28	1276.05	1283.27
2.	03jan2001	1347.76	1274.62	1347.56
3.	04jan2001	1350.24	1329.14	1333.34
4.	05jan2001	1334.77	1294.95	1298.35
5.	08jan2001	1298.35	1276.29	1295.86

We will use the first 37 observations from these data:

```
. twoway rcap high low date in 1/37
```




Advanced use

twoway rcap works well when combined with a horizontal line representing a base value:

```
. use https://www.stata-press.com/data/r18/sp500, clear
(S&P 500)
. generate month = month(date)
. sort month
. by month: egen lo = min(volume)
. by month: egen hi = max(volume)
. format lo hi %10.0gc
. summarize volume
```

Variable	Obs	Mean	Std. dev.	Min	Max
volume	248	12320.68	2585.929	4103	23308.3


```
. by month: keep if _n==_N
(236 observations deleted)
. twoway rcap lo hi month,
  xlabel(1 "J" 2 "F" 3 "M" 4 "A" 5 "M" 6 "J"
        7 "J" 8 "A" 9 "S" 10 "O" 11 "N" 12 "D")
  xtitle("Month of 2001")
  ytitle("High and Low Volume")
  yaxaxis(1 2)
  ylabel(12321 "12,321 (mean)", axis(2) angle(0))
  ytitle("", axis(2))
  yline(12321, lstyle(foreground))
  msize(*2)
  title("Volume of the S& P 500", margin(b+2.5))
  note("Source: Yahoo!Finance and Commodity Systems Inc.")
```



Advanced use 2

twoway rcap also works well when combined with a scatterplot to produce hi-lo-middle graphs. Returning to the first 37 observations of the S&P 500 used in the first example, we add a scatterplot of the closing value:

```
. use https://www.stata-press.com/data/r18/sp500, clear
(S&P 500)
. keep in 1/37
(211 observations deleted)
. twoway rcap high low date || scatter close date
```



Also see

- [G-2] [graph twoway rarea](#) — Range plot with area shading
- [G-2] [graph twoway rbar](#) — Range plot with bars
- [G-2] [graph twoway rcapsym](#) — Range plot with spikes capped with marker symbols
- [G-2] [graph twoway rconnected](#) — Range plot with connected lines⁺
- [G-2] [graph twoway rline](#) — Range plot with lines⁺
- [G-2] [graph twoway rscatter](#) — Range plot with markers
- [G-2] [graph twoway rspike](#) — Range plot with spikes

Stata, Stata Press, and Mata are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. StataNow and NetCourseNow are trademarks of StataCorp LLC. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2023 StataCorp LLC, College Station, TX, USA. All rights reserved.



For suggested citations, see the FAQ on [citing Stata documentation](#).